
1 Corrections required

1.1 f_8_2_4_4_EUTRA
	Template name
	f_TC_8_2_4_4_EUTRA

	Reason for change
	According to the 36.523-1 Table 8.2.4.4.3.2-1, Step 1 in the rrcConnectionReconfiguration message, In the IE MobilityControlInfo the IE carrierFreq should be set as not present.

Current TTCN implementation does not set this IE accordingly, and needs to be corrected.

	Summary of change
	Instead of calling function f_EUTRA_RRC_ConnReconfigHO_Common, called f_EUTRA_RRC_ConnReconfigHO_IntraLTE_Security.

As a result of this, removed some code from TTCN that is highlighted in green

	TTCN module
	RRC_Handover.ttcn

	MCC160 Comment
	


Before Change:
	function f_TC_8_2_4_4_EUTRA ( ) runs on EUTRA_PTC
  {
    var SubFrameTiming_Type v_Timing;
    var SubFrameTiming_Type v_Timing_Release;
    var SubFrameTiming_Type v_Timing_Setup;
    var SubFrameTiming_Type v_Timing_T304;
    var PhysCellId v_PhysicalCellIdentity_Cell1;
    var ShortMAC_I v_ShortMAC_I;
    var EUTRA_SecurityParams_Type v_Auth_Params;
    var UInt_Type v_T304_IntegerType;
    var float v_T311 := 10.0; //T311 is 10 seconds according to 36.508 cl. 4.4.3.3 Table 4.4.3.3-1;
    var float v_T304 := 1.0;  //1 second = 1000ms, according to 36.508 cl. 4.6.5 Table 4.6.5-1
    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;
    var NextHopChainingCount v_NextHopChainingCount := 2; //@sic R5s110356 & R5-113104 sic@
    var template (omit) RRCConnectionReconfiguration_v890_IEs v_RRCConnectionReconfiguration_v890_IEs := omit; //@sic R5s190055, R5-192719 sic@
    timer t_WaitRRCConnRestbReq;
    f_EUTRA_Init ( c1 );
    v_T304 := f_EUTRA_SetTimerToleranceMax( eutra_Cell1, rrcTimer, v_T304 );
    v_T304_IntegerType := float2int(v_T304 * 1000.0); //@sic R5s110776 change 2.1 sic@
    v_T311 := f_EUTRA_SetTimerToleranceMax( eutra_Cell1, rrcTimer, v_T311 );
    v_PhysicalCellIdentity_Cell1 := f_EUTRA_CellInfo_GetPhyCellId ( eutra_Cell1 );
    //Create and configure cell 1
    f_EUTRA_CellConfig_Def ( eutra_Cell1 );
    //Bring UE to initial state
    f_EUTRA_Preamble ( eutra_Cell1, STATE2_IDLEUPDATE );
    f_EUTRA_RbEst_Def ( eutra_Cell1 );
    v_Auth_Params := f_EUTRA_Security_Get ( );
    f_EUTRA_TestBody_Set ( true );
    //Configure SS to not send response on RACH preamble reception on Cell 1
    //This configuration represents Table 8.2.4.4.3.2-2: Parallel behaviour
    f_EUTRA_SS_ConfigRachProcedure ( eutra_Cell1,
                                     omit,
                                     cs_RachProcedureConfig_NoResponse (f_EUTRA_CellInfo_GetDL_ChBandwidth(eutra_Cell1)));
    //@siclog "Step 1" siclog@
    //Send RRCConnectionReconfiguration to UE for security reconfiguration on Cell 1
    v_Timing := f_EUTRA_GetNextSendOccasion ( eutra_Cell1, tsc_EUTRA_DelayBeforeIntraCellHO); //@sic R5-106675 & R5s100714 sic@
    v_RRCConnectionReconfiguration_v890_IEs := f_EUTRA_CellInfo_GetReconfig_BR(eutra_Cell1); //@sic R5s190055, R5-192719 sic@
    //@sic R5s120036 sic@
    f_EUTRA_RRC_ConnReconfigHO_Common ( eutra_Cell1,
                                        omit,
                                        f_Generate_cs_MobilityControlInfo_HO(eutra_Cell1),
                                        f_Generate_cs_508_RadioResourceConfigDedicated_HO(eutra_Cell1),
                                        cs_SecurityConfigHO_IntraLTE(-, false, v_NextHopChainingCount),
                                        cs_TimingInfo_EUTRA_NB(v_Timing),
                                        v_RRCConnectionReconfiguration_v890_IEs); //@sic R5s190055, R5-192719 sic@
    v_Timing_Release := f_EUTRA_TimingInfoAdd(v_Timing, 5);   // add 5ms
    //Release SRBs and DRBs
    f_EUTRA_SS_SRBs_DRBs_Release(eutra_Cell1, cs_TimingInfo_EUTRA_NB(v_Timing_Release));
    //Setup SRBs and DRBs again
    v_Timing_Setup := f_EUTRA_TimingInfoAdd(v_Timing_Release, 5);   // add 5ms
    f_EUTRA_SS_SRBs_DRBs_Config(eutra_Cell1, cs_TimingInfo_EUTRA_NB(v_Timing_Setup));
    // Activate RRC security at SS target cell (Cell1) for SRB 1, 2 and 1 AM DRB
    //* Security configuration is not changed here because UE will have to revert back to the configuration
    //* used in the source cell upon T304 expiry (36.331 cl. 5.3.5.6).
    //v_Auth_Params := f_EUTRA_SS_AS_ActivateSecurity_HO_Reest ( eutra_Cell1, v_Auth_Params, v_NextHopChainingCount );
    //f_EUTRA_Security_Set ( v_Auth_Params );
    //@siclog "Step 2" siclog@
    //SS does not respond to UE during MAC Random Access Preamble procedure till T304 expires
    //No response is taken care of by earlier RACH configuration
    //Normal response when T304 expires is taken care of by below configuration.
    //Configure Cell 1 for normal PRACH operation
    v_Timing_T304 := f_EUTRA_TimingInfoAdd(v_Timing, v_T304_IntegerType);   // add T304 in milliseconds
    f_EUTRA_SS_ConfigRachProcedure_Def(eutra_Cell1, tsc_C_RNTI_Def, -, -, -, cs_TimingInfo_EUTRA_NB(v_Timing_T304));  /* @sic acc. to R5s100029 change 3 sic@ */
    //Calculate ShortMAC-I to be received in RRCConnectionReestablishment Request from UE
    //* shortMAC-I is calculated with the KRRCint key and integrity protection algorithm that was used in the source cell
    //* since no security configuration was done after sending RRCConnectionReconfiguration, the shortMAC-I can be
    //* calculated here.
    v_ShortMAC_I := f_EUTRA_Calculate_ShortMAC ( eutra_Cell1, eutra_Cell1 );
    //Start Watchdog timer
    t_WaitRRCConnRestbReq.start(v_T311);
    //@siclog "Step 3" siclog@
    //Receive RRCConnectionReestablishment Request
    /* 36.331 cl. 5.3.7.4
     * The UE shall set the contents of RRCConnectionReestablishmentRequest message as follows:
     *  1>  set the ue-Identity as follows:
     *   2> set the c-RNTI to the C-RNTI used in the source cell (handover and mobility from E-UTRA failure)
     *      or used in the cell in which the trigger for the re-establishment occurred (other cases)
    */
    SRB.receive ( car_SRB0_RrcPdu_IND ( eutra_Cell1,
                                        cr_508_RRCConnectionReestablishmentRequest ( tsc_C_RNTI_Def,
                                                                                     v_PhysicalCellIdentity_Cell1,
                                                                                     cr_ReestablishmentCause_HandoverFailure,
                                                                                     v_ShortMAC_I ) ) );
    t_WaitRRCConnRestbReq.stop;
    f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Test Case 8.2.4.4 Step 3");
    //Activate RRC security at SS target cell (Cell1) for SRB 1, 2 and 1 AM DRB based on current Kenb and NH parameter
    //@sic R5s100714 : Reactivate AS security before sending RRCConnectionReestablishment sic@
    v_NextHopChainingCount := 0;    // Default configuration value according to TS 36.508 Table 4.6.1-10
    v_Auth_Params := f_EUTRA_SS_AS_ActivateSecurity_HO_Reest ( eutra_Cell1, v_Auth_Params, v_NextHopChainingCount );
    f_EUTRA_Security_Set ( v_Auth_Params );
    //@siclog "Step 4" siclog@
    //Transmit RRCConnectionReestablishment to resume SRB1 operation and re-activate security
    f_EUTRA_RRC_RRCConnectionReestablishment_Def ( eutra_Cell1, v_RRC_TI, v_NextHopChainingCount );
    //@siclog "Step 5" siclog@
    //Receive RRCConnectionReestablishmentComplete
    SRB.receive ( car_SRB1_RrcPdu_IND ( eutra_Cell1,  cr_508_RRCConnectionReestablishmentComplete( v_RRC_TI ) ) );
    //@siclog "Step 6 - 7" siclog@
    //RRCConnectionReconfiguration to resume SRB2 and existing DRBs
    f_EUTRA_RRC_RRCConnectionReconfiguration_Resume_SRB2_DRBs ( eutra_Cell1, v_RRC_TI );
    //@siclog "Step 8" siclog@
    //Check if UE is in state RRC_Connected
    f_EUTRA_508Check_ConnectedState ( eutra_Cell1 );
    f_EUTRA_TestBody_Set ( false );
    //Switch/power off UE
    f_EUTRA_Postamble ( eutra_Cell1, E2_CONNECTED );
  }


After Change:

	function f_TC_8_2_4_4_EUTRA ( ) runs on EUTRA_PTC
  {
    var SubFrameTiming_Type v_Timing;
    var SubFrameTiming_Type v_Timing_Release;
    var SubFrameTiming_Type v_Timing_Setup;
    var SubFrameTiming_Type v_Timing_T304;
    var PhysCellId v_PhysicalCellIdentity_Cell1;
    var ShortMAC_I v_ShortMAC_I;
    var EUTRA_SecurityParams_Type v_Auth_Params;
    var UInt_Type v_T304_IntegerType;
    var float v_T311 := 10.0; //T311 is 10 seconds according to 36.508 cl. 4.4.3.3 Table 4.4.3.3-1;
    var float v_T304 := 1.0;  //1 second = 1000ms, according to 36.508 cl. 4.6.5 Table 4.6.5-1
    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;
    var NextHopChainingCount v_NextHopChainingCount := 2; //@sic R5s110356 & R5-113104 sic@
    timer t_WaitRRCConnRestbReq;
    f_EUTRA_Init ( c1 );
    v_T304 := f_EUTRA_SetTimerToleranceMax( eutra_Cell1, rrcTimer, v_T304 );
    v_T304_IntegerType := float2int(v_T304 * 1000.0); //@sic R5s110776 change 2.1 sic@
    v_T311 := f_EUTRA_SetTimerToleranceMax( eutra_Cell1, rrcTimer, v_T311 );
    v_PhysicalCellIdentity_Cell1 := f_EUTRA_CellInfo_GetPhyCellId ( eutra_Cell1 );
    //Create and configure cell 1
    f_EUTRA_CellConfig_Def ( eutra_Cell1 );
    //Bring UE to initial state
    f_EUTRA_Preamble ( eutra_Cell1, STATE2_IDLEUPDATE );
    f_EUTRA_RbEst_Def ( eutra_Cell1 );
    v_Auth_Params := f_EUTRA_Security_Get ( );
    f_EUTRA_TestBody_Set ( true );
    //Configure SS to not send response on RACH preamble reception on Cell 1
    //This configuration represents Table 8.2.4.4.3.2-2: Parallel behaviour
    f_EUTRA_SS_ConfigRachProcedure ( eutra_Cell1,
                                     omit,
                                     cs_RachProcedureConfig_NoResponse (f_EUTRA_CellInfo_GetDL_ChBandwidth(eutra_Cell1)));
    //@siclog "Step 1" siclog@
    //Send RRCConnectionReconfiguration to UE for security reconfiguration on Cell 1
    v_Timing := f_EUTRA_GetNextSendOccasion ( eutra_Cell1, tsc_EUTRA_DelayBeforeIntraCellHO); //@sic R5-106675 & R5s100714 sic@
    //@sic R5s120036 sic@
    f_EUTRA_RRC_ConnReconfigHO_IntraLTE_Security (eutra_Cell1,
                                                  eutra_Cell1,
                                                  cs_SecurityConfigHO_IntraLTE(-, false, v_NextHopChainingCount),
                                                  cs_TimingInfo_EUTRA_NB(v_Timing));
    v_Timing_Release := f_EUTRA_TimingInfoAdd(v_Timing, 5);   // add 5ms
    //Release SRBs and DRBs
    f_EUTRA_SS_SRBs_DRBs_Release(eutra_Cell1, cs_TimingInfo_EUTRA_NB(v_Timing_Release));
    //Setup SRBs and DRBs again
    v_Timing_Setup := f_EUTRA_TimingInfoAdd(v_Timing_Release, 5);   // add 5ms
    f_EUTRA_SS_SRBs_DRBs_Config(eutra_Cell1, cs_TimingInfo_EUTRA_NB(v_Timing_Setup));
    // Activate RRC security at SS target cell (Cell1) for SRB 1, 2 and 1 AM DRB
    //* Security configuration is not changed here because UE will have to revert back to the configuration
    //* used in the source cell upon T304 expiry (36.331 cl. 5.3.5.6).
    //v_Auth_Params := f_EUTRA_SS_AS_ActivateSecurity_HO_Reest ( eutra_Cell1, v_Auth_Params, v_NextHopChainingCount );
    //f_EUTRA_Security_Set ( v_Auth_Params );
    //@siclog "Step 2" siclog@
    //SS does not respond to UE during MAC Random Access Preamble procedure till T304 expires
    //No response is taken care of by earlier RACH configuration
    //Normal response when T304 expires is taken care of by below configuration.
    //Configure Cell 1 for normal PRACH operation
    v_Timing_T304 := f_EUTRA_TimingInfoAdd(v_Timing, v_T304_IntegerType);   // add T304 in milliseconds
    f_EUTRA_SS_ConfigRachProcedure_Def(eutra_Cell1, tsc_C_RNTI_Def, -, -, -, cs_TimingInfo_EUTRA_NB(v_Timing_T304));  /* @sic acc. to R5s100029 change 3 sic@ */
    //Calculate ShortMAC-I to be received in RRCConnectionReestablishment Request from UE
    //* shortMAC-I is calculated with the KRRCint key and integrity protection algorithm that was used in the source cell
    //* since no security configuration was done after sending RRCConnectionReconfiguration, the shortMAC-I can be
    //* calculated here.
    v_ShortMAC_I := f_EUTRA_Calculate_ShortMAC ( eutra_Cell1, eutra_Cell1 );
    //Start Watchdog timer
    t_WaitRRCConnRestbReq.start(v_T311);
    //@siclog "Step 3" siclog@
    //Receive RRCConnectionReestablishment Request
    /* 36.331 cl. 5.3.7.4
     * The UE shall set the contents of RRCConnectionReestablishmentRequest message as follows:
     *  1>  set the ue-Identity as follows:
     *   2> set the c-RNTI to the C-RNTI used in the source cell (handover and mobility from E-UTRA failure)
     *      or used in the cell in which the trigger for the re-establishment occurred (other cases)
    */
    SRB.receive ( car_SRB0_RrcPdu_IND ( eutra_Cell1,
                                        cr_508_RRCConnectionReestablishmentRequest ( tsc_C_RNTI_Def,
                                                                                     v_PhysicalCellIdentity_Cell1,
                                                                                     cr_ReestablishmentCause_HandoverFailure,
                                                                                     v_ShortMAC_I ) ) );
    t_WaitRRCConnRestbReq.stop;
    f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Test Case 8.2.4.4 Step 3");
    //Activate RRC security at SS target cell (Cell1) for SRB 1, 2 and 1 AM DRB based on current Kenb and NH parameter
    //@sic R5s100714 : Reactivate AS security before sending RRCConnectionReestablishment sic@
    v_NextHopChainingCount := 0;    // Default configuration value according to TS 36.508 Table 4.6.1-10
    v_Auth_Params := f_EUTRA_SS_AS_ActivateSecurity_HO_Reest ( eutra_Cell1, v_Auth_Params, v_NextHopChainingCount );
    f_EUTRA_Security_Set ( v_Auth_Params );
    //@siclog "Step 4" siclog@
    //Transmit RRCConnectionReestablishment to resume SRB1 operation and re-activate security
    f_EUTRA_RRC_RRCConnectionReestablishment_Def ( eutra_Cell1, v_RRC_TI, v_NextHopChainingCount );
    //@siclog "Step 5" siclog@
    //Receive RRCConnectionReestablishmentComplete
    SRB.receive ( car_SRB1_RrcPdu_IND ( eutra_Cell1,  cr_508_RRCConnectionReestablishmentComplete( v_RRC_TI ) ) );
    //@siclog "Step 6 - 7" siclog@
    //RRCConnectionReconfiguration to resume SRB2 and existing DRBs
    f_EUTRA_RRC_RRCConnectionReconfiguration_Resume_SRB2_DRBs ( eutra_Cell1, v_RRC_TI );
    //@siclog "Step 8" siclog@
    //Check if UE is in state RRC_Connected
    f_EUTRA_508Check_ConnectedState ( eutra_Cell1 );
    f_EUTRA_TestBody_Set ( false );
    //Switch/power off UE
    f_EUTRA_Postamble ( eutra_Cell1, E2_CONNECTED );
  }


