
1. Overview

This document lists all the essencial changes needed to correct problems in the TTCN implementation of test case 8.1.1.4.1 which is part of the NR5GC test suite in the ‘iwd-TTCN3-B2019-06_D19wk38’ ATS delivery. 

With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence.

3. Verification Test Summary

Test Case:
8.1.1.4.1
Test Group:
\NR5GC\8_1_1\
ATS Version:
iwd-TTCN3-B2019-06_D19wk38
System Simulator used:
Keysight 5G Protocol Conformance Toolset
UE used:
HiSilicon Balong 5000
Verification Status:
PASS


4. Corrections required for NR5GC TC 8.1.1.4.1 on wk38 ATS
4.1 Change 1

	Function name
	f_TC_8_1_1_4_1_NR5GC

	Reason for change
	1. Cell 11 has a different tracking area than cell 1 ( as specified in the prose) but this configuration would cause the UE to initiate a Tracking Area Update procedure after step 18 which is not currently handled.

2. Test loop mode should be deactivated before postamble

3. Postamble will not happen on cell 1.


	Summary of change
	1. Use NR cell 2 instead of NR cell 11

Note : A prose CR associated with this change will be raised at RAN5#85

2. Deactivate test loop mode after test case body

3. Use cell 2 for postamble



	TTCN module
	RRC_Resume_NR5GC

	MCC160 Comment
	


Before 

function f_TC_8_1_1_4_1_NR5GC() runs on NR5GC_PTC

  { // RRC resume / Suspend-Resume / RNA update / Success

    //@sic R5-197053 sic@

    f_NR5GC_Init(NR_2);

    //Create and configure cell

    f_NR_CellConfig_Def(nr_Cell1);

    f_NR_CellConfig_Def(nr_Cell11);

    f_NR5GC_Preamble(nr_Cell1, STATE_INACTIVE_2A, PING_Or_TEST_LOOPModeB_ON);

    f_NR_SetCellPower(nr_Cell11, tsc_NR_SuitableIntraFreqCellSSS_EPRE_FR1, tsc_NR_SuitableIntraFreqCellSSS_EPRE_FR2);

    f_NR_TestBody_Set(true);

    fl_TC_8_1_1_4_1_TestBody();

    f_NR_TestBody_Set(false);

    f_NR_Postamble(nr_Cell1, STATE_CONNECTED_3A);
  }

After change

function f_TC_8_1_1_4_1_NR5GC() runs on NR5GC_PTC

  { // RRC resume / Suspend-Resume / RNA update / Success

    //@sic R5-197053 sic@

    f_NR5GC_Init(NR_2);

    //Create and configure cell

    f_NR_CellConfig_Def(nr_Cell1);

    f_NR_CellConfig_Def(nr_Cell2);

    f_NR5GC_Preamble(nr_Cell1, STATE_INACTIVE_2A, PING_Or_TEST_LOOPModeB_ON);

    f_NR_SetCellPower(nr_Cell11, tsc_NR_SuitableIntraFreqCellSSS_EPRE_FR1, tsc_NR_SuitableIntraFreqCellSSS_EPRE_FR2);

    f_NR_TestBody_Set(true);

    fl_TC_8_1_1_4_1_TestBody();

    f_NR_TestBody_Set(false);

f_NR5GC_OpenUE_TestLoopMode_Deactivate_TestMode(nr_Cell2);
    f_NR_Postamble(nr_Cell2, STATE_CONNECTED_3A);
  }

4.2 Change 2

	Function name
	f_NR_SendRRCResume_Def

	Reason for change
	Currently in the function f_NR_SendRRCResume_Def(), existing DRBs are reconfigured in the SS on RRC connection resume but AS security is not reactivated.

	Summary of change
	Activate AS security to start iontegrity protection and ciphering for the reconfigured DRBs

	TTCN module
	NR5GC_RRCSteps

	MCC160 Comment
	


Before 

function f_NR_SendRRCResume_Def(NR_CellId_Type p_CellId,

                             template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,

                             RRC_TransactionIdentifier p_RRC_TI := tsc_NR_RRC_TI_Def) runs on NR5GC_PTC

  {

    var NR_DRBInfoList_Type v_ExistingDRBs := f_NR5GC_MobileInfo_GetDRBInfoList();

    var template (value) NR_RadioBearerList_Type v_SS_Config := f_NR5GC_GetRadioBearerList(v_ExistingDRBs);

    var template (value) NR_RachProcedureConfig_Type v_NR_RachProcedureConfig := f_NR_InitialiseRachProcedure(p_CellId);

    // switch on UL grant assignments

    f_NR_ULGrantConfiguration_Start(p_CellId);

    // refresh security keys and activate security => moved to the function bringing UE to Inactive state to give time to SS to configure security parameters

    // @sic R5s190505 change 1 sic@

    SRB.send(cas_NR_SRB1_RrcPdu_REQ(p_CellId, p_TimingInfo, cs_38508_RRCResume(p_RRC_TI)));

    //Reconfigure DRB in SS - existing DRBs are not reconfigured during release

    f_NR_SS_CommonRadioBearerConfig(p_CellId, v_SS_Config, p_TimingInfo);

    f_NR_SS_RachProcedureConfig(p_CellId, v_NR_RachProcedureConfig);

  }

After change

function f_NR_SendRRCResume_Def(NR_CellId_Type p_CellId,

                             template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,

                             RRC_TransactionIdentifier p_RRC_TI := tsc_NR_RRC_TI_Def) runs on NR5GC_PTC

  {

    var NR_DRBInfoList_Type v_ExistingDRBs := f_NR5GC_MobileInfo_GetDRBInfoList();

    var template (value) NR_RadioBearerList_Type v_SS_Config := f_NR5GC_GetRadioBearerList(v_ExistingDRBs);

    var template (value) NR_RachProcedureConfig_Type v_NR_RachProcedureConfig := f_NR_InitialiseRachProcedure(p_CellId);

    var NR_SecurityParams_Type v_SecurityParams := f_NR_Security_Get(); 
    // switch on UL grant assignments

    f_NR_ULGrantConfiguration_Start(p_CellId);

    // refresh security keys and activate security => moved to the function bringing UE to Inactive state to give time to SS to configure security parameters

    // @sic R5s190505 change 1 sic@

    SRB.send(cas_NR_SRB1_RrcPdu_REQ(p_CellId, p_TimingInfo, cs_38508_RRCResume(p_RRC_TI)));

    //Reconfigure DRB in SS - existing DRBs are not reconfigured during release

    f_NR_SS_CommonRadioBearerConfig(p_CellId, v_SS_Config, p_TimingInfo);

    v_SecurityParams.AS_Ciphering.ActTimeList := f_NR_RRC_CipherActTime_GetInitial(p_CellId);

    f_NR_SS_RRC_EnableIntProt_CiphULandDL(p_CellId,

                                          v_SecurityParams.AS_Integrity,

                                          v_SecurityParams.AS_Ciphering);

    f_NR_Security_Set(v_SecurityParams);
    f_NR_SS_RachProcedureConfig(p_CellId, v_NR_RachProcedureConfig);

  }

4.3 Change 3
	Function name
	a_AspFromSystemAdaptorHandler_NRNG

	Reason for change
	RRCResumeRequest1 at step 14 will use “ccch1” branch, not currently handled in this altstep

	Summary of change
	Add handling for ccch1 messages

	TTCN module
	NasEmu5G_Main_NRNG

	MCC160 Comment
	


Before 

altstep a_AspFromSystemAdaptorHandler_NRNG(inout NG_NasSecurity_Type p_NG_NasSecurityByRef) runs on NASEMU_NR_BASE_PTC

  {

    var NR_RRC_PDU_IND v_AspFromSYS;

    var NR_RRC_MSG_Indication_Type v_RrcIndication;

    var UL_DCCH_Message v_RrcPdu;

    var ULInformationTransfer_IEs v_ULInformationTransfer;

    var template (value) NR_SRB_COMMON_IND v_SRB_COMMON_IND;

    var template (value) NG_NasEmu_DecodingInfo_Type v_DecodingInfo;

    var DedicatedNAS_Message v_NAS_DedicatedMsg[maxDRB];   // even though there is always only one we use an array

    var integer v_NoOfNasPdus;

    var boolean v_RrcPduPresent;

    var boolean v_IntegrityError;

    var integer I;

    [] SYS_SRB.receive(car_NRNG_PduFromSYS) -> value v_AspFromSYS {

      v_RrcIndication := v_AspFromSYS.RrcPdu;

      v_SRB_COMMON_IND := cas_NRNG_SignallingAspToTC(v_AspFromSYS.Common);

      if (ischosen(v_RrcIndication.Ccch)) {                                     /* CCCH -> SRB0 */

        v_SRB_COMMON_IND.Signalling.Rrc := v_RrcIndication;

      } else {                                                                  /* DCCH -> SRB1/2/3 */

        v_RrcPdu := v_RrcIndication.Dcch;

... ...

After change

altstep a_AspFromSystemAdaptorHandler_NRNG(inout NG_NasSecurity_Type p_NG_NasSecurityByRef) runs on NASEMU_NR_BASE_PTC

  {

    var NR_RRC_PDU_IND v_AspFromSYS;

    var NR_RRC_MSG_Indication_Type v_RrcIndication;

    var UL_DCCH_Message v_RrcPdu;

    var ULInformationTransfer_IEs v_ULInformationTransfer;

    var template (value) NR_SRB_COMMON_IND v_SRB_COMMON_IND;

    var template (value) NG_NasEmu_DecodingInfo_Type v_DecodingInfo;

    var DedicatedNAS_Message v_NAS_DedicatedMsg[maxDRB];   // even though there is always only one we use an array

    var integer v_NoOfNasPdus;

    var boolean v_RrcPduPresent;

    var boolean v_IntegrityError;

    var integer I;

    [] SYS_SRB.receive(car_NRNG_PduFromSYS) -> value v_AspFromSYS {

      v_RrcIndication := v_AspFromSYS.RrcPdu;

      v_SRB_COMMON_IND := cas_NRNG_SignallingAspToTC(v_AspFromSYS.Common);

      if (ischosen(v_RrcIndication.Ccch) or ischosen(v_RrcIndication.Ccch1)) {                                     /* CCCH -> SRB0 */

        v_SRB_COMMON_IND.Signalling.Rrc := v_RrcIndication;

      }

      else {                                                                  /* DCCH -> SRB1/2/3 */

        v_RrcPdu := v_RrcIndication.Dcch;

... ...

4.4 Change 4
	Function name
	F_NR_RRCResumeComplete_Def()

	Reason for change
	Current implementation will give a run-time error if RRCResumeComplete is recioeved without a NAS message

	Summary of change
	Check for presence of NAS message

	TTCN module
	NR5GC_RRCSteps.ttcn

	MCC160 Comment
	


Before 

function f_NR_RRCResumeComplete_Def(NR_CellId_Type p_CellId,

                                     template(present) UL_DCCH_Message p_ExpectedRRCResumeComplete := cr_38508_RRCResumeComplete)

    runs on NR_BASE_PTC return  template (omit) NG_NAS_MSG_Indication_Type

  {

    var template (omit) NG_NAS_MSG_Indication_Type v_DedicatedNAS_Message;

    var NR_SRB_COMMON_IND v_ReceivedAsp;

    //The UE transmits an RRCResumeComplete message.

    SRB.receive(car_NR_SRB1_RrcPdu_IND(p_CellId, p_ExpectedRRCResumeComplete)) -> value v_ReceivedAsp;

    v_DedicatedNAS_Message := v_ReceivedAsp.Signalling.Nas[0];

    return v_DedicatedNAS_Message;

  }
After change

function f_NR_RRCResumeComplete_Def(NR_CellId_Type p_CellId,

                                     template(present) UL_DCCH_Message p_ExpectedRRCResumeComplete := cr_38508_RRCResumeComplete)

    runs on NR_BASE_PTC return  template (omit) NG_NAS_MSG_Indication_Type

  {

    var template (omit) NG_NAS_MSG_Indication_Type v_DedicatedNAS_Message := omit;

    var NR_SRB_COMMON_IND v_ReceivedAsp;

    //The UE transmits an RRCResumeComplete message.

    SRB.receive(car_NR_SRB1_RrcPdu_IND(p_CellId, p_ExpectedRRCResumeComplete)) -> value v_ReceivedAsp;
    if (ispresent (v_ReceivedAsp.Signalling.Nas))

   {
          v_DedicatedNAS_Message := v_ReceivedAsp.Signalling.Nas[0];
   }
    return v_DedicatedNAS_Message;

  }

4.5 Change 5

	Function name
	fl_TC_8_1_1_4_1_TestBody

	Reason for change
	1. Cell 11 has a different tracking area than cell 1 ( as specified in the prose) but this configuration would cause the UE to initiate a Tracking Area Update procedure after step 18 which is not currently handled.
2. The function f_NR_CheckDataPath() is currently not applicable to NR5GC as it deals with PDCP SDUs and does not take SDAP layer into account.
3. Since the UE moves to cell 2 following step 15 it is not required to reactivate security inside function f_NR_RRCInactiveDef(). Security should be activated separately for cell 2 instead.
4. In step16 when resuming RRC connection on a different cell, UE will continue with existing RLC sequence numbering for SRB2 and DRB unless explicitly told to restablish.

	Summary of change
	1. Use NR cell 2 instead of NR cell 11
2. Use function f_NR5GC_CehckDataPath() as defined in MCC160 comments to CR R5s190936.
3. Activate security on cell 2.
4. Include masterCellGroup with rlc-BearerToAddModList indicating reestablishRlc for SRB2 and DRB.
Note : An prose CR associated with changes 1 and 4 will be raised at RAN5#85

	TTCN module
	RRC_Resume_NR5GC

	MCC160 Comment
	


Before 

function fl_TC_8_1_1_4_1_TestBody() runs on NR5GC_PTC

  {

    //@sic R5-197053 sic@

    var NR_DRBInfoList_Type v_ExistingDRBs;

    var integer v_NoOfDRBs := f_NR5GC_MobileInfo_GetNoOfDRBs();

    var NR_DRBInfo_Type v_DRBInfo;

    var integer v_DRBId;

    var integer v_Index := -1;

    var integer v_ModificationPeriodFrames;

    var float v_ModificationPeriod;

    timer t_WaitTime := 60.0;

    //@siclog "Step 1" siclog@

    //The SS transmits a Paging message including a matched identity (correct fullI-RNTI).

    f_NR_Paging_Def(nr_Cell1, RRC_INACTIVE);

    //@siclog "Step 2" siclog@

    //Check: Does the UE transmit an RRCResumeRequest message by setting resumeIdentity to the stored shortI-RNTI value?

    f_NR_RRC_ResumeRequest_Def(nr_Cell1, tsc_NR_ShortI_RNTI_Value1, mt_Access);

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 2");

    //@siclog "Step 3" siclog@

    //The SS transmits an RRCResume message.

    f_NR_SendRRCResume_Def(nr_Cell1);

    //@siclog "Step 4" siclog@

    //The UE transmits an RRCResumeComplete message.

    f_NR_RRCResumeComplete_Def(nr_Cell1);
    //@siclog "Step 5" siclog@

    //Check: Does the test result of generic test procedure in TS 38.508-1 subclause 4.9.1 indicate that the UE is capable of exchanging IP data on DRB#n associated with default Internet PDU session?

    if (v_NoOfDRBs == 0) {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "No DRBs configured");

    }

    //Get DRB Info for DRB associated with 1st PDU session

    v_DRBId := f_NR_GetDefaultDRB_ForFirstPDUSession();

    v_ExistingDRBs := f_NR5GC_MobileInfo_GetDRBInfoList();

    v_Index := f_NR_DRBInfo_FindEntry(v_ExistingDRBs, v_DRBId);

    v_DRBInfo := v_ExistingDRBs[v_Index];

    f_NR_CheckDataPath(nr_Cell1, PDN_2, dedicatedBearer, v_DRBInfo.DRB_Id, cs_RlcBearerRouting_NR(nr_Cell1), -, "Step 5");
    //@siclog "Step 6" siclog@

    //The SS transmits an RRCRelease message including both fullI-RNTI and shortI-RNTI in suspendConfig.

    f_NR_RRC_InactiveDef(nr_Cell1);

    //@siclog "Step 7" siclog@

    //The SS changes the SIB1 of NR Cell 1 to set the useFullResumeID to True.

    f_NR_CellInfo_SetSIB1_UseFullResumeID(nr_Cell1, true_);

    //@siclog "Step 8" siclog@

    //The SS transmits a Short message on PDCCH using P-RNTI indicating a systemInfoModification.

    f_NR_ModifySysinfo(nr_Cell1, true, RRC_INACTIVE);

    //@siclog "Step 9" siclog@

    //Wait for 2.1* modification period second for the UE to receive new system information.

    v_ModificationPeriodFrames := f_NR_CalculateModificationPeriod(nr_Cell1);

    v_ModificationPeriod := int2float(v_ModificationPeriodFrames)/100.0; //Modification period in seconds

    f_Delay(2.1 * v_ModificationPeriod);

    //@siclog "Step 10" siclog@

    //The SS transmits a Paging message including a matched identity (correct fullI-RNTI).

    f_NR_Paging_Def(nr_Cell1, RRC_INACTIVE);

    //@siclog "Step 11" siclog@

    //Check: Does the UE transmit an RRCResumeRequest1 message by setting resumeIdentity to the stored fullI-RNTI value?

    SRB.receive(car_NR_SRB0_RrcPdu_IND1(nr_Cell1, cr_38508_RRCResumeRequest1(tsc_NR_I_RNTI_Value1)));

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 10");

    //@siclog "Step 12" siclog@

    //The SS transmits an RRCResume message.

    f_NR_SendRRCResume_Def(nr_Cell1);

    //@siclog "Step 13" siclog@

    //The UE transmits an RRCResumeComplete message.

    f_NR_RRCResumeComplete_Def(nr_Cell1);
    //@siclog "Step 14" siclog@

    //The UE transmits an RRCResumeComplete message.

    f_NR_RRC_InactiveDef(nr_Cell1);
    //@siclog "Step 15" siclog@

    //The SS changes NR Cell 1 to Non-suitable "Off" cell according to 38.508-1 [4] Table 6.2.2.1-3.

    f_NR_SetCellPower(nr_Cell1,  tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE);

    //@siclog "Step 16" siclog@

    //Check: Does the UE transmit an RRCResumeRequest message with resumeCause set to rna-Update on NR Cell 11 within 60s?

    t_WaitTime.start;

    alt {

      [] SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell11, cr_38508_RRCResumeRequest(tsc_NR_ShortI_RNTI_Value1, rna_Update)))

      {

         t_WaitTime.stop;

         f_NR_PreliminaryPass (__FILE__, __LINE__, "Test Case 8.1.1.4.1 Step 3");

      }

      [] t_WaitTime.timeout

      {

         f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Test Case 8.1.1.4.1 Step 3");

      }

    }

    //@siclog "Step 17" siclog@

    //The SS transmits an RRCResume message

    f_NR_SendRRCResume_Def(nr_Cell2);
    //@siclog "Step 18" siclog@

    //The UE transmit an RRCResumeComplete message to complete the resume procedure

    f_NR_RRCResumeComplete_Def(nr_Cell1);
    //@siclog "Step 19" siclog@

    //Check: Does the test result of generic test procedure in TS 38.508-1 subclause 4.9.1 indicate that the UE is capable of exchanging IP data on DRB#n associated with default Internet PDU session?

    f_NR_CheckDataPath(nr_Cell1, PDN_2, dedicatedBearer, v_DRBInfo.DRB_Id, cs_RlcBearerRouting_NR(nr_Cell11), -, "Step 19");
  }

After change

function fl_TC_8_1_1_4_1_TestBody() runs on NR5GC_PTC

  {

    //@sic R5-197053 sic@

    var NR_DRBInfoList_Type v_ExistingDRBs;

    var integer v_NoOfDRBs := f_NR5GC_MobileInfo_GetNoOfDRBs();

    var NR_DRBInfo_Type v_DRBInfo;

    var integer v_DRBId;

    var integer v_Index := -1;

    var integer v_ModificationPeriodFrames;

    var float v_ModificationPeriod;

    timer t_WaitTime := 60.0;

    var template (value) CellGroupConfig v_CellGroupConfig;

    var template(value) NR_RLC_BearerToAddModList_Type v_NR_RLC_BearerToAddModList;             

    //@siclog "Step 1" siclog@

    //The SS transmits a Paging message including a matched identity (correct fullI-RNTI).

    f_NR_Paging_Def(nr_Cell1, RRC_INACTIVE);

    //@siclog "Step 2" siclog@

    //Check: Does the UE transmit an RRCResumeRequest message by setting resumeIdentity to the stored shortI-RNTI value?

    f_NR_RRC_ResumeRequest_Def(nr_Cell1, tsc_NR_ShortI_RNTI_Value1, mt_Access);

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 2");

    //@siclog "Step 3" siclog@

    //The SS transmits an RRCResume message.

    f_NR_SendRRCResume_Def(nr_Cell1);

    //@siclog "Step 4" siclog@

    //The UE transmits an RRCResumeComplete message.    

     f_NR_RRCResumeComplete_Def(nr_Cell1);
    //@siclog "Step 5" siclog@

    //Check: Does the test result of generic test procedure in TS 38.508-1 subclause 4.9.1 indicate that the UE is capable of exchanging IP data on DRB#n associated with default Internet PDU session?

    if (v_NoOfDRBs == 0) {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "No DRBs configured");

    }

    //Get DRB Info for DRB associated with 1st PDU session

    v_DRBId := f_NR_GetDefaultDRB_ForFirstPDUSession();

    v_ExistingDRBs := f_NR5GC_MobileInfo_GetDRBInfoList();

    v_Index := f_NR_DRBInfo_FindEntry(v_ExistingDRBs, v_DRBId);

    v_DRBInfo := v_ExistingDRBs[v_Index];

    f_NR5GC_CheckDataPath(nr_Cell1, PDN_2, dedicatedBearer, v_DRBInfo.DRB_Id, "Step 5");
    //@siclog "Step 6" siclog@

    //The SS transmits an RRCRelease message including both fullI-RNTI and shortI-RNTI in suspendConfig.

    f_NR_RRC_InactiveDef(nr_Cell1);

    //@siclog "Step 7" siclog@

    //The SS changes the SIB1 of NR Cell 1 to set the useFullResumeID to True.

    f_NR_CellInfo_SetSIB1_UseFullResumeID(nr_Cell1, true_);

    //@siclog "Step 8" siclog@

    //The SS transmits a Short message on PDCCH using P-RNTI indicating a systemInfoModification.

    f_NR_ModifySysinfo(nr_Cell1, true, RRC_INACTIVE);

    //@siclog "Step 9" siclog@

    //Wait for 2.1* modification period second for the UE to receive new system information.

    v_ModificationPeriodFrames := f_NR_CalculateModificationPeriod(nr_Cell1);

    v_ModificationPeriod := int2float(v_ModificationPeriodFrames)/100.0; //Modification period in seconds

    f_Delay(2.1 * v_ModificationPeriod);

    //@siclog "Step 10" siclog@

    //The SS transmits a Paging message including a matched identity (correct fullI-RNTI).

    f_NR_Paging_Def(nr_Cell1, RRC_INACTIVE);

    //@siclog "Step 11" siclog@

    //Check: Does the UE transmit an RRCResumeRequest1 message by setting resumeIdentity to the stored fullI-RNTI value?

    SRB.receive(car_NR_SRB0_RrcPdu_IND1(nr_Cell1, cr_38508_RRCResumeRequest1(tsc_NR_I_RNTI_Value1)));

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 10");

    //@siclog "Step 12" siclog@

    //The SS transmits an RRCResume message.

    f_NR_SendRRCResume_Def(nr_Cell1);

    //@siclog "Step 13" siclog@

    //The UE transmits an RRCResumeComplete message.    

     f_NR_RRCResumeComplete_Def(nr_Cell1);
    //@siclog "Step 14" siclog@

    f_NR_RRC_InactiveDef(nr_Cell1, -, false); 
    f_NR_SS_AS_ActivateSecurity_Current(nr_Cell2 ,tsc_NR_38508_NextHopChainingCount);
    //@siclog "Step 15" siclog@

    //The SS changes NR Cell 1 to Non-suitable "Off" cell according to 38.508-1 [4] Table 6.2.2.1-3.

    f_NR_SetCellPower(nr_Cell1,  tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE);

    //@siclog "Step 16" siclog@

    //Check: Does the UE transmit an RRCResumeRequest message with resumeCause set to rna-Update on NR Cell 11 within 60s?

    t_WaitTime.start;

    alt {

      [] SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell2, cr_38508_RRCResumeRequest(tsc_NR_ShortI_RNTI_Value1, rna_Update)))

      {

         t_WaitTime.stop;

         f_NR_PreliminaryPass (__FILE__, __LINE__, "Test Case 8.1.1.4.1 Step 3");

      }

      [] t_WaitTime.timeout

      {

         f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Test Case 8.1.1.4.1 Step 3");

      }

    }

    //@siclog "Step 17" siclog@

    //The SS transmits an RRCResume message   

    // Build RLC-BearerConfig for SRB2 and DRB with reestablishRlc set to true

    v_NR_RLC_BearerToAddModList[0] := cs_NR_BearerConfigDef(f_NR_LogicalChannelId_DRB(v_DRBInfo.DRB_Id), omit, omit, omit, true_);

    v_NR_RLC_BearerToAddModList[1] := cs_NR_BearerConfigDef(f_NR_LogicalChannelId_SRB(tsc_SRB2), omit, omit, omit, true_);

    v_CellGroupConfig := cs_38508_CellGroupConfig(tsc_NR_CellGroupId_MCG, v_NR_RLC_BearerToAddModList);

     f_NR_SendRRCResume_Def(nr_Cell2, -, -, v_CellGroupConfig);
    //@siclog "Step 18" siclog@

    //The UE transmit an RRCResumeComplete message to complete the resume procedure        

    f_NR_RRCResumeComplete_Def(nr_Cell1);
    //@siclog "Step 19" siclog@
    //Check: Does the test result of generic test procedure in TS 38.508-1 subclause 4.9.1 indicate that the UE is capable of exchanging IP data on DRB#n associated with default Internet PDU session?                       

    f_NR5GC_CheckDataPath(nr_Cell2, PDN_2, dedicatedBearer, v_DRBInfo.DRB_Id, "Step 19");
  }
4.4 Change 4

	Function name
	cs_38508_RRCResume_CellGroupConfig

	Reason for change
	In step16 when resuming RRC connection on a different cell, UE will continue with existing RLC sequence numbering for SRB2 and DRB unless explicitly told to reestablish.



	Summary of change
	Add parameter to include masterCellGroup in RRCResume


	TTCN module
	RRC_Resume_NR5GC

	MCC160 Comment
	


Before

template(value) DL_DCCH_Message cs_38508_RRCResume(RRC_TransactionIdentifier p_RRC_TI := tsc_NR_RRC_TI_Def):=

  {  // 38.508-1 Table 4.6.1-17: RRCResume

     /* @status    APPROVED (NR5GC) */

    message_ := {

      c1 := {

        rrcResume := {

          rrc_TransactionIdentifier := p_RRC_TI,

          criticalExtensions := {

            rrcResume := {

              radioBearerConfig           := omit,

              masterCellGroup             := omit,

              measConfig                  := omit,

              fullConfig                  := omit,

              lateNonCriticalExtension    := omit,

              nonCriticalExtension        := omit

            }

          }

        }

      }

    }

  };
After

template(value) DL_DCCH_Message cs_38508_RRCResume(RRC_TransactionIdentifier p_RRC_TI := tsc_NR_RRC_TI_Def, template (value) CellGroupConfig p_CellGroupConfig := omit):=

  {  // 38.508-1 Table 4.6.1-17: RRCResume

     /* @status    APPROVED (NR5GC) */

    message_ := {

      c1 := {

        rrcResume := {

          rrc_TransactionIdentifier := p_RRC_TI,

          criticalExtensions := {

            rrcResume := {

              radioBearerConfig           := omit,

              masterCellGroup             := p_CellGroupConfig,

              measConfig                  := omit,

              fullConfig                  := omit,

              lateNonCriticalExtension    := omit,

              nonCriticalExtension        := omit

            }

          }

        }

      }

    }

  };
4.4 Change 4

	Function name
	cs_38508_RRCResume_CellGroupConfig

	Reason for change
	In step16 when resuming RRC connection on a different cell, UE will continue with existing RLC sequence numbering for SRB2 and DRB unless explicitly told to reestablish.



	Summary of change
	Add parameter to include masterCellGroup in RRCResume
Note : A prose CR associated with this change will be raised at RAN5#85



	TTCN module
	RRC_Resume_NR5GC

	MCC160 Comment
	


Before

function f_NR_SendRRCResume_Def(NR_CellId_Type p_CellId,

                             template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,

                             RRC_TransactionIdentifier p_RRC_TI := tsc_NR_RRC_TI_Def) runs on NR5GC_PTC

  {

    var NR_DRBInfoList_Type v_ExistingDRBs := f_NR5GC_MobileInfo_GetDRBInfoList();

    var template (value) NR_RadioBearerList_Type v_SS_Config := f_NR5GC_GetRadioBearerList(v_ExistingDRBs);

    var template (value) NR_RachProcedureConfig_Type v_NR_RachProcedureConfig := f_NR_InitialiseRachProcedure(p_CellId);

    // switch on UL grant assignments

    f_NR_ULGrantConfiguration_Start(p_CellId);

    // refresh security keys and activate security => moved to the function bringing UE to Inactive state to give time to SS to configure security parameters

    // @sic R5s190505 change 1 sic@

    SRB.send(cas_NR_SRB1_RrcPdu_REQ(p_CellId, p_TimingInfo, cs_38508_RRCResume(p_RRC_TI)));

    //Reconfigure DRB in SS - existing DRBs are not reconfigured during release

    f_NR_SS_CommonRadioBearerConfig(p_CellId, v_SS_Config, p_TimingInfo);

    f_NR_SS_RachProcedureConfig(p_CellId, v_NR_RachProcedureConfig);

  }
After

function f_NR_SendRRCResume_Def(NR_CellId_Type p_CellId,

                             template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,

                             RRC_TransactionIdentifier p_RRC_TI := tsc_NR_RRC_TI_Def, template (value) CellGroupConfig p_CellGroupConfig := omit) runs on NR5GC_PTC

  {

    var NR_DRBInfoList_Type v_ExistingDRBs := f_NR5GC_MobileInfo_GetDRBInfoList();

    var template (value) NR_RadioBearerList_Type v_SS_Config := f_NR5GC_GetRadioBearerList(v_ExistingDRBs);

    var template (value) NR_RachProcedureConfig_Type v_NR_RachProcedureConfig := f_NR_InitialiseRachProcedure(p_CellId);

    // switch on UL grant assignments

    f_NR_ULGrantConfiguration_Start(p_CellId);

    // refresh security keys and activate security => moved to the function bringing UE to Inactive state to give time to SS to configure security parameters

    // @sic R5s190505 change 1 sic@

    SRB.send(cas_NR_SRB1_RrcPdu_REQ(p_CellId, p_TimingInfo, cs_38508_RRCResume(p_RRC_TI, p_CellGroupConfig)));

    //Reconfigure DRB in SS - existing DRBs are not reconfigured during release

    f_NR_SS_CommonRadioBearerConfig(p_CellId, v_SS_Config, p_TimingInfo);

    f_NR_SS_RachProcedureConfig(p_CellId, v_NR_RachProcedureConfig);

  }
5. Execution Log Files

5.1 HiSilicon Balong 5000
The HiSilicon Balong 5000UE passed this test case on Keysight 5G Protocol Conformance Toolset Solution on NR5GC band n41.The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file: 
TC_8_1_1_4_1_Log.html
· 

(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)

In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

